
Gestión de arrays en el entorno NATURAL 
 
 
El disponer de un ejemplo completo y claro favorece que el programador 
utilice mejor todos los recursos que pone a su disposición el sistema. Tal 
es el objetivo buscado con el presente artículo, el cual, tal y como indi-
ca su título, presenta un ejemplo muy completo sobre el tratamiento de las 
tablas o arrays, así como del tratamiento de fechas en el entorno NATURAL. 
 
1.- Introducción:  
 
NATURAL realiza la gestión de tablas o arrays en los programas de una for-
ma ágil y cómoda, y lo mismo ocurre con la información que se puede obte-
ner de las variables de tipo fecha (DATE) tal y como se muestra en el 
ejemplo que acompaña al articulo. 
 
El hecho de trabajar en grandes sistemas con terminales 3270, que son de 
solo texto, hace difícil mostrar todo un año con el formato de “calendario 
tradicional” en una única pantalla. Por esta razón solo se sacarán semes-
tres, mientras que en el listado, se sacará el año completo. 
 
2.- Tablas usadas 
 
La figura 1 muestra el calendario del año 1997; mientras que la figura 2 
ayuda a identificar las 5 tablas que se usan en el mapa {CALENDAM} y que 
se comentan a continuación:  
 

 
 
T-DIA-MES  (N2/1:12,1:6,1:7) Tabla para contener cada día de cada uno de 
los meses: Numérico de 2 dígitos, y tres dimensiones (12 meses; 6 semanas 
por mes/7 días por semana) 
 
T-LIT-MES (A12/1:12) : Tabla para contener los nombres de los doce meses. 
 
Para contener la cabecera del día de la semana, se recurre a una tabla re-
definida. (ver líneas 350-400 del listado de CALENDAP). Realmente lo que 
se hace es reservar un espacio de 42 bytes de memoria (T-LIT-DIA-SEM). Ese 
espacio, se redefine como una tabla de 7 posiciones y 6 ocurrencias, que 

solo sirve para facilitar la carga de los literales (T-LIT-DIA). Y, por 
ultimo, ese mismo espacio de 42 posiciones, se redefine como una tabla de 
3 dimensiones (T-DIA-L: 2 filas * 3 meses * 7 días por mes):  
             
T-NUM-SEMANA (n2/1:12,1:6): Esta tabla contendrá el número de semana (di-
vida en 12 meses y hasta 6 semanas por mes)  
 
T-DIA-CV (c/1:12,1:6,1:7): Esta tabla se define para contener variables de 
control, a fin de que los días festivos aparezcan con mas brillo (intensi-
ficados).  
 
El resto de variables pueden verse en los listados de los fuentes que se 
acompañan, tanto de CALENDAM y CALENDAP 
 

 
 

  
3.- Comentarios al programa: 
 
La figura 3 muestra la cabecera del programa, donde cada línea tiene el 
siguiente significado:  
 

 
 

0520: Carga todas las ocurrencias 6 y 7 de todos los meses y de todas las 
semanas de la tabla de variables de control con el atributo de intensifi-
cado. 
0530: Activa la tecla PF3 y la asigna el comando “MENU” 
0540: Crea una ventana de 50*12, y sitúa el ángulo superior izquierdo en 
la línea 5 columna 20. 
0550: Carga la variable W-FECHA-N con el valor numérico de la fecha ac-
tual, para que en la ventana ya salga cargado el dato del año a listar con 
el valor del año actual. 

0560-0650: Petición al usuario de año que se desea listar, depositando es-
te valor en la variable W-ANIO, que a su vez es un subcampo de W-FECHA-N. 
El parámetro *OUTIN que precede a la variable es para especificar que di-
cho campo es de salida (OUT) y entrada (IN), ya que en caso de no especi-
ficarle asume que es *IN (solo entrada)   
0620: Dentro del mismo input se pide al usuario que elija si desea la sa-
lida mediante mapa o mediante listado realizado con write. 
0670: Restaura los atributos de pantalla a los valores por defecto. Es de-
cir cierra la ventana.  
 
La figura 4 muestra un fragmento de programa correspondiente a la elabora-
ción de datos, según el esquema propuesto en los artículos de esta sección 
(Solo Programadores num. 24 y 25) 
  

 
 

0750: Establece como lenguaje del sistema el Castellano, para que los va-
lores devueltos por el sistema vengan en este lenguaje. 
0760: Inicializa los valores de día y mes a 1 para obtener, junto al año 
que ha especificado el usuario, el primer día del año.  
0790: Mueve el contenido del campo numérico que contiene la fecha (primer 
día del año), a una variable de tipo ‘date’ utilizando la sentencia MOVE 
EDITED. Esta sentencia, por regla general, cuando se usa con variables de 
tipo Date(D), usa como variable receptora o emisora una variable de tipo 
alfabética, de ahí que se utilice la variable W-FECHA-A que es la redefi-
nición de W-FECHA-N (numérica).  
0800: Obtiene el año romano de la fecha.  
0810: Obtiene el día de la semana de la fecha contenida en la variable de 
tipo date.  
0830: Pasa a mayúsculas el día de la semana obtenido en 820, ya que el 
sistema deja este valor con el primer carácter en mayúsculas y el resto en 
minúsculas, y para la siguiente validación se necesita que todo este en 
mayúsculas, a menos que cuando se está editando el programa, se haya espe-
cificado el comando de terminal %L para que no se pasen a mayúsculas los 
valores que se escriban en las líneas 850-910. 
0840-0930: Carga la variable auxiliar w-aux con el numero del día de la 
semana.  
  
La siguientes instrucciones no tienen mayor complejidad, por lo que pueden 
seguirse en el listado del programa que se acompaña en el disco; o la fi-
gura y. 

0950-0990: se corresponden con la carga de los nombres de los días de la 
semana moviendo el literal ‘LMXJVSD’   
1030-1050: Estructura repetitiva para tratar mediante FOR cada uno de los 
12 meses de un año. 
 

 
 

La figura 5 muestra,  quizás, la parte más complicada del programa, pues 
carga los datos variables correspondientes a un mes. 
Los índices que se usan para las distintas repetitivas tipo FOR, son los 
subcampos W-MES y W-DIA redefinidos de la variable W-FECHA-N.  
La variable auxiliar W-AUX se utiliza para ir arrastrando el día dentro de 
la semana de un mes a otro, por eso, cada vez que se carga un día, se suma 
1 a esa variable, y cuando supera el valor 7 actualiza los demás índices.  
 
1190 - 1200: Obtiene el nombre del mes desde el sistema.  
1210-1220: Obtiene el numero de la semana del primer día del mes 
1270-1460: repetitiva tipo FOR para cargar los 31 días del mes. 
 
 
 
4.- Comentarios al mapa:  
 
Los comentarios asociados al mapa, se tienen que realizar sobre los hard-
copys que muestran el interface proporcionado por NATURAL para el trata-
miento de los mapas, ya que este entorno no admite otra forma de defini-
ción de mapas, a no ser por mapping directo en el propio programa.  
  
La figura 6 muestra las variables usadas en el mapa CALENDAM. Es de resal-
tar que las tablas T-NUM-SEMANA y T-DIA-MES se encuentran repetidas, pues 
se tratan por trimestre. 

La figura 7 muestra el detalle correspondiente a la tabla T-LIT-MES. Esta 
tabla, de 1 dimensión con 12 ocurrencias, es tratada como una tabla de 
3*2. Para lo cual, a la izquierda de INDEX HORIZONTAL se pone un 1 por ser 
la primera dimensión y querer que las ocurrencias aparezcan en horizontal, 
pero solo tres ocurrencias, y además comenzando por el valor contenido en 
la variables W-IND-1. Al colocar un 2 en INDEX VERTICAL hace que se listen 
las siguientes ocurrencias como si fueran las dos dimensiones citadas. En 
resumen, solo se listan las 6 ocurrencias que se necesitan.  
 

 
 
La figura 8 muestra el detalle de la definición de la tabla T-NUM-SEMANA. 
Esta tabla, de 2 dimensiones (12*6), es tratada como si fuera de 3*6, don-
de el primer índice establece el trimestre, pues presenta 3 ocurrencias 
comenzando por la especificada en la variable W-IND-1, y el segundo, el 
número de semana dentro del mes.  

La figura 9 muestra la definición mas compleja ya que hace uso de las tres 
dimensiones para representar la tabla T-DIA-L. Esta tabla, como puede ver-
se en la primera línea, es una tabla de 3 dimensiones, concretamente de 
2*3*7 ocurrencias. Donde: 
-la primera dimensión, (con un “1” bajo el titulo “Dimensions”, con 2 ocu-
rrencias, especifica el trimestre (por decirlo de alguna forma) ya que se 
corresponde con el índice vertical (Espaciado entre ocurrencias: 10 lí-
neas).  
-la segunda dimensión, con 3 ocurrencias, corresponde con cada uno de los 
meses dentro del trimestre (espaciado entre ocurrencias: 7 columnas según 
se aprecia en la figura); y, por último,  

-la tercera dimensión establece las 7 ocurrencias de cada mes, escritas  
en sentido horizontal, por eso la opción (H) en el campo “INDEX”.  
dejando entre ocurrencia y ocurrencia 2 columnas.  
 
 
La figura 10 muestra la definición de la  tabla que se corresponde con los 
días de cada mes, siendo tratada (“pinchada”) por trimestre, por eso se 
encuentra repetida su definición en el listado de variables del mapa (fi-
gura 6). Concretamente, la figura muestra la definición correspondiente al 
segundo trimestre del semestre que se muestre, iniciándose éste por el mes 
especificado en la variable W-IND-2. La primera dimensión establece el 
trimestre. La segunda, establece la semana (en sentido vertical), y la 
tercera, se escribe en horizontal, por eso el valor (H) en el campo INDEX, 
y tiene 7 ocurrencias.  
 

 
 
Por último, la figura 11 muestra la definición de la variable W-IND-1, 
utilizada como índice de alguna tabla del mapa. Este campo realmente solo 
sirve para que aparezca definido en el área de parámetros del objeto mapa 
(CALENDAM), ya que en el caso de no hacerlo así, se produciría un error de 
parámetros al ser invocado desde el programa. Así pues, como no tiene otra 
función, se le asigna el atributo (AD=N) de no visualizable (Non-Display), 
y una longitud de numérico (NL=1) de 1 dígito, a pesar de estar definida 
como (N03) numérico de 3 dígitos.  

5.- Sentencias asociadas con el tratamiento de tablas:  
 
NATURAL también dispone de otras sentencias para la definición y gestión 
de las tablas. Este apartado, pretende recordar aquellas instrucciones que 
sin estar incluidas en el ejemplo, están asociadas con el tema tratado.  
 
Así, en el caso de querer saber cual es la suma de todas las ocurrencias 
de los días de un mes, solo habría que hacer:  
W-Suma = + t-dia-mes (1,*,*) 
 
Otra situación muy normal en cualquier aplicación consiste en seleccionar 
una ocurrencia, por ejemplo una provincia, de entre las que encuentran en 
una lista. Para ello, se suele utilizar otra tabla auxiliar para que el 
usuario marque en ella, la provincia elegida.  
Las sentencias EXAMINE y IF SELECTION evitan el tener que generar una es-
tructura repetitiva, tipo FOR por ejemplo, para tratar esa tabla auxiliar 
y saber si se ha seleccionado mas de una ocurrencia. Ejemplos: 
IF SELECTION NOT UNIQUE REINPUT ‘==> error...’ 
EXAMINE T-MARCA (*) FOR ‘x’ GIVING INDEX W-INDICE 

6.- Caso particular: 
 
NATURAL permite pasar datos de un programa a otro vía STACK, siendo reco-
gidos estos valores en el programa llamado mediante la sentencia INPUT. 
(Solo Programadores Julio-Agos-Sept-96).  
Pues bien, cuando la variable receptora es mayor de 80 posiciones, el sis-
tema devuelve un error diciendo que dicho valor no cabe en una línea. Este 
error se solventa utilizando en el INPUT la redefinición en forma de tabla 
de una variable de longitud inferior, tal y como se muestra a continua-
ción.  
 

DEFINE DATA LOCAL  
01 W-ENTRADA (A150) 
01 REDEFINE W-ENTRADA 
02 W-CAMPO (A50/1:3) 
... 
INPUT W-CAMPO (1:3) 

 
7.- Utilidad:  
 
Este mes, la utilidad que se acompaña, es la que ha servido de base para 
el artículo. No obstante, en el fichero JMP_TAB.NAT se incluye otro pro-
grama NATURAL que muestra otra forma de tratar los campos redefinidos de 
una tabla, así como el resultado de su ejecución. 
 
Por ultimo, desde estas líneas se lanza a los lectores una invitación para 
publicar las utilidades desarrolladas por ellos mismos en NATURAL. A tal 
efecto, deben hacer llegar las mismas,  bien a la redacción de la revista, 
o a la dirección del autor del articulo <jmpeco@luznet.es>. Así mismo, 
también se admiten direcciones de foros y WEBs relacionadas con el entorno 
del lenguaje NATURAL (Ejemplo: http:/www.sag.es). 

1         LIBRERIA ......: UTILITYS     PROGRAMA... ......: CALENDAP                      FECHA : 30/10/94 
          USUARIO .......: JMPDES                                                         HORA  : 17:06:22 
          TERMINAL ......: EMSID046 MVS/ESA IMS/DC                                        PAGINA:     7 
          MODIFICADO POR : EMSID046 EL DIA : 30/10/1994 A LAS  17:01:26 
          ------------------------------------------------------------------------------------------------- 
0                       0010   ************************************************************************ 
                        0020   * 
                        0030   * APLICACION : UTILIDAD 
                        0040   * PROGRAMA   : CALENDAP 
                        0050   * OBJETIVO   : LISTAR EL CALENDARIO DE UN A¤O 
                        0060   * AUTOR      : JMP 
                        0070   * FECHA      : 23/08/94 
                        0080   * 
                        0090   ************************************************************************ 
                        0100   * 
                        0120   LOCAL 
                        0130      1 W-SEMANA           (N02) 
                        0140      1 W-NUM-SEMANA-A     (A02) 
                        0150      1 REDEFINE W-NUM-SEMANA-A 
                        0160        2 W-NUM-SEMANA-N   (N02) 
                        0170   * 
                        0180      1 W-FECHA-N          (N8) 
                        0190      1 REDEFINE W-FECHA-N 
                        0200        2 W-ANIO           (N4) 
                        0210        2 W-MES            (N2) 
                        0220        2 W-DIA            (N2) 
                        0230      1 REDEFINE W-FECHA-N 
                        0240        2 W-FECHA-A        (A8) 
                        0250   * 
                        0260      1 W-DATE             (D) 
                        0270      1 W-ANIO-ROMANO      (A13) 
                        0280      1 W-LIT-DIA1         (A10) 
                        0290      1 W-AUX              (N3) 
                        0300      1 X                  (N3) 
                        0310      1 W-SALIDA           (A1)  INIT <'M'> 
                        0320      1 T-AUX              (N2/1:7) 
                        0330      1 W-IND-1            (N3) 
                        0340      1 W-IND-2            (N3) 
                        0350   * 
                        0360      1 T-LIT-DIA-SEM             (A42) 
                        0370      1 REDEFINE T-LIT-DIA-SEM 
                        0380        2 T-DIA-L                 (A1/1:2,1:3,1:7) 
                        0390      1 REDEFINE T-LIT-DIA-SEM 
                        0400        2 T-DIA-SEMANA            (A7/1:6) 
                        0410   * 
                        0420      1 T-DIA-MES          (N02/1:12,1:6,1:7) 
                        0430      1 T-DIA-CV           (C/1:12,1:6,1:7) 
                        0440      1 T-LIT-MES          (A12/1:12) 
                        0450      1 T-NUM-SEMANA       (N02/1:12,1:6) 
                        0460   * 
                        0470   END-DEFINE 
                        0480   ************************************************************************ 
                        0490   * 
                        0500   * -------------- ENTRADA DE DATOS 
                        0510   * 
                        0520       ASSIGN T-DIA-CV(*,*,6:7) = (AD=I) 
                        0530       SET KEY PF3='MENU' 
                        0540       SET CONTROL 'WFL50C12B5/20' 
                        0550       MOVE *DATN TO W-FECHA-N 
           INPUT        0560       INPUT (SG=OFF) 
                        0570               ' *******************************************' 
                        0580           /   ' *               CALENDARIO                *' 
                        0590           /   ' *******************************************' 
                        0600           /   ' *                                         *' 
                        0610           /   ' *          A¤O :' *OUTIN W-ANIO '                    *' 
                        0620           /   ' *       SALIDA :' *OUTIN W-SALIDA 
                        0630                                   ' M:MAPA L:LISTADO      *' 
                        0640           /   ' *                                         *' 
                        0650           /   ' *******************************************' 
                        0660   * 
                        0670       SET CONTROL 'WB' 
           1IF          0680       IF *PF-KEY = 'PF3' 
                        0690       OR W-ANIO  = 0 
                        0700          STOP 
           1END-IF      0710       END-IF 
                        0720   * 
                        0730   *  ------------------------------- DETERMINAR DIA DE LA SEMANA 
                        0740   * 

1         LIBRERIA ......: UTILITYS     PROGRAMA... ......: CALENDAP                      FECHA : 30/10/94 
          USUARIO .......: JMPDES                                                         HORA  : 17:06:22 
          TERMINAL ......: EMSID046 MVS/ESA IMS/DC                                        PAGINA:     8 
          MODIFICADO POR : EMSID046 EL DIA : 30/10/1994 A LAS  17:01:26 
          ----------------------------------------------------------------------------------------------------------- 
0                       0750      MOVE  4 TO *LANGUAGE 
                        0760      MOVE 01 TO W-MES 
                        0770      MOVE 01 TO W-DIA 
                        0780   * 
                        0790      MOVE EDITED W-FECHA-A            TO W-DATE (EM=YYYYMMDD) 
                        0800      MOVE EDITED W-DATE    (EM=R)     TO W-ANIO-ROMANO 
                        0810      MOVE EDITED W-DATE    (EM=N(10)) TO W-LIT-DIA1 
                        0820   * 
                        0830      EXAMINE W-LIT-DIA1 TRANSLATE INTO UPPER CASE 
           DECIDE       0840      DECIDE ON FIRST VALUE OF W-LIT-DIA1 
                        0850             VALUE 'LUNES'     MOVE 1 TO W-AUX 
                        0860             VALUE 'MARTES'    MOVE 2 TO W-AUX 
                        0870             VALUE 'MIERCOLES' MOVE 3 TO W-AUX 
                        0880             VALUE 'JUEVES'    MOVE 4 TO W-AUX 
                        0890             VALUE 'VIERNES'   MOVE 5 TO W-AUX 
                        0900             VALUE 'SABADO'    MOVE 6 TO W-AUX 
                        0910             VALUE 'DOMINGO'   MOVE 7 TO W-AUX 
                        0920             NONE VALUE IGNORE 
           END-DECIDE   0930      END-DECIDE 
                        0940   * 
                        0950   * -------------- CARGAR TABLA DE NOMBRES DE DIA DE SEMANA 
                        0960   * 
                        0970      FOR X = 1 TO 6 
                        0980          MOVE 'LMXJVSD' TO T-DIA-SEMANA (X) 
                        0990      END-FOR 
                        1000   * 
                        1010   *  ------------------------------- CARGAR VALORES VARIABLES 
                        1020   * 
                        1030      FOR W-MES = 1 TO 12 
           PERFORM      1040          PERFORM CARGAR-MES 
                        1050      END-FOR 
                        1060   * 
           1IF          1070      IF W-SALIDA = 'M' 
           PERFORM      1080         THEN PERFORM SACAR-SEMESTRES 
           1 ELSE       1090         ELSE PERFORM LISTAR-RESULTADOS 
           1END-IF      1100      END-IF 
                        1110   * 
           **FETCH      1120      FETCH 'MENU' 
                        1130   * 
                        1140   ************************************************************************ 
                        1150   DEFINE SUBROUTINE CARGAR-MES 
                        1160   * 
                        1170          MOVE 1 TO W-SEMANA 
                        1180          MOVE 1 TO W-DIA 
                        1190          MOVE EDITED W-FECHA-A         TO W-DATE (EM=YYYYMMDD) 
                        1200          MOVE EDITED W-DATE (EM=L(12)) TO T-LIT-MES (W-MES) 
                        1210          MOVE EDITED W-DATE (EM=ZW)    TO W-NUM-SEMANA-A 
                        1220          MOVE W-NUM-SEMANA-N           TO T-NUM-SEMANA (W-MES,1) 
           1IF          1230          IF W-MES = 1 
                        1240             MOVE 0 TO W-NUM-SEMANA-N 
           1END-IF      1250          END-IF 
                        1260   * 
                        1270          FOR W-DIA = 1 TO 31 
           1IF          1280              IF W-FECHA-N NE MASK (YYYYMMDD) 
           ESCAPE       1290                 ESCAPE BOTTOM 
           1END-IF      1300              END-IF 
 
                        ... 
 
 
                        1380              MOVE W-DIA TO T-DIA-MES (W-MES,W-SEMANA,W-AUX) 
                        1390              ADD 1 TO W-AUX 
           1IF          1400              IF W-AUX GT 7 
                        1410                 MOVE 1 TO W-AUX 
                        1420                 ADD  1 TO W-SEMANA 
                        1430                 ADD  1 TO W-NUM-SEMANA-N 
                        1440                 MOVE W-NUM-SEMANA-N  TO T-NUM-SEMANA (W-MES,W-SEMANA) 
           1END-IF      1450              END-IF 
                        1460          END-FOR 
                        1470   * 
           1IF          1480          IF W-AUX = 1 

1         LIBRERIA ......: UTILITYS     PROGRAMA... ......: CALENDAP                      FECHA : 30/10/94 
          USUARIO .......: JMPDES                                                         HORA  : 17:06:22 
          TERMINAL ......: EMSID046 MVS/ESA IMS/DC                                        PAGINA:     9 
          MODIFICADO POR : EMSID046 EL DIA : 30/10/1994 A LAS  17:01:26 
          ------------------------------------------------------------------------------------------------ 
0                       1490             MOVE 0 TO T-NUM-SEMANA (W-MES,W-SEMANA) 
           1END-IF      1500          END-IF 
                        1510   * 
                        1520   END-SUBROUTINE 
                        1530   ************************************************************************ 
                        1540   DEFINE SUBROUTINE SACAR-SEMESTRES 
                        1550   * 
                        1560      FOR X = 1 TO 2 
                        1570          ASSIGN W-IND-1 = ( X - 1 ) * 6 + 1 
                        1580          ASSIGN W-IND-2 = W-IND-1 + 3 
           INPUT        1590          INPUT USING MAP 'CALENDAM' 
                        1600      END-FOR 
                        1610   * 
           INPUT        1620   *  INPUT USING MAP 'CALENDAM' T-DIA-CV  (1:06,*,*) 
                        1630   *                             T-DIA-L   (1:2,1:3,1:7) 
                        1640   *                             T-DIA-MES (1:06,*,*) 
                        1650   *                             T-LIT-MES (1:06) 
                        1660   *                             T-NUM-SEMANA (1:06,1:6) 
                        1670   *                             W-ANIO 
                        1680   *                             W-ANIO-ROMANO 
                        1690   *                             W-LIT-DIA1 
                        1700   * 
           INPUT        1710   *  INPUT USING MAP 'CALENDAM' T-DIA-CV(7:12,*,*) 
                        1720   *                             T-DIA-L   (1:2,1:3,1:7) 
                        1730   *                             T-DIA-MES (7:12,*,*) 
                        1740   *                             T-LIT-MES (7:12) 
                        1750   *                             T-NUM-SEMANA (7:12,1:6) 
                        1760   *                             W-ANIO 
                        1770   *                             W-ANIO-ROMANO 
                        1780   *                             W-LIT-DIA1 
                        1790   * 
           MAP          1800   WRITE    USING MAP 'CALENDAW' T-DIA-CV     (7:12,*,*) 
                        1810                                 T-DIA-L      (01:2,1:3,1:7) 
                        1820                                 T-DIA-MES    (07:12,*,*) 
                        1830                                 T-LIT-MES    (07:12) 
                        1840                                 T-NUM-SEMANA (07:12,1:6) 
                        1850                                 W-ANIO 
                        1860                                 W-ANIO-ROMANO 
                        1870                                 W-LIT-DIA1 
                        1880   _ 
           MAP          1890   *  MAP USING MAP 'JMPCALEM' T-DIA-L(*,*,*) T-DIA(7:12,*,*) T-MES(7:12) 
                        1900   *                             ANIO W-NOM-DIA 
                        1910   * 
                        1920   END-SUBROUTINE 
                        1930   ************************************************************************ 
                        1940   DEFINE SUBROUTINE LISTAR-RESULTADOS 
                        1950   * 
                        1960      FOR W-MES = 1 TO 12 
                        1970          FOR W-SEMANA = 1 TO 6 
                        1980              FOR W-DIA = 1 TO 7 
                        1990                  ASSIGN  T-AUX (W-DIA) = T-DIA-MES (W-MES,W-SEMANA,W-DIA) 
                        2000              END-FOR 
                        2010              WRITE T-AUX (*) (EM=ZZ) 
                        2020          END-FOR 
                        2030          WRITE ' ' 
                        2040      END-FOR 
                        2050   * 
                        2060   END-SUBROUTINE 
                        2070   ************************************************************************ 
                        2080   * 
                        2090      END 
                        2100   * 

1         LIBRERIA ......: UTILITYS     MAPA ..........: CALENDAM                         FECHA : 30/10/94 
          USUARIO .......: JMPDES                                                         HORA  : 17:06:23 
          TERMINAL ......: EMSID101                                                       PAGINA:    11 
          MODIFICADO POR : EMSID101 EL DIA :  30/10/1994 A LAS  17:21:32 
          ------------------------------------------------------------------------------------------------ 
- 
0         FORMATO :  FORMAT PS=023 LS=080 ZP=OFF SG=OFF KD=OFF IP=OFF 
          TIPO ...:  MAPA INVOCADO CON INPUT 
- 
-         *********************************************************************************** 
          | Calendario correspondiente al a¤o: 9999 o XXXXXXXXXXXXX  comienza en XXXXXXXXXX | 
          |                                                                                 | 
          |         XXXXXXXXXXXX              XXXXXXXXXXXX              XXXXXXXXXXXX        | 
          |                                                                                 | 
          |       X  X  X  X  X  X  X       X  X  X  X  X  X  X       X  X  X  X  X  X  X   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |                                                                                 | 
          |                                                                                 | 
          |        XXXXXXXXXXXX              XXXXXXXXXXXX              XXXXXXXXXXXX         | 
          |                                                                                 | 
          |       X  X  X  X  X  X  X       X  X  X  X  X  X  X       X  X  X  X  X  X  X   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   99 ZZ ZZ ZZ ZZ ZZ ZZ ZZ   | 
          |                                                                                 | 
          |                                                                                 | 
          *********************************************************************************** 
- 
-               ------------------------------------------------------------ 
            1   T-DIA-CV                                         C /01:06,01:06,01 
            2  T-DIA-SEMANA                                      A 01/01:02,01:03, 
            3  T-DIA-VALOR                                       N 02,0/01:06,01:0 
            4  T-MES-LITERAL                                     A 012/01:06 
            5  T-NUM-SEMANA                                      N 02,0/01:06,01:0 
            6  W-ANIO                                            N 04,0 
            7  W-ANIO-ROMANO-XX                                  A 013 
            8  W-LIT-DIA-SEMANA                                  A 010 
                ------------------------------------------------------------ 
-         0645 * MAP2: VALIDATION ***************************************************** 
          0646 * MAP2: END OF MAP ***************************************************** 
          0647 END 

****** ****************************** top of data ***************************** 
000010 ************************************************************************ 
000020 * 
000030 * APLICACION : UTILIDAD 
000040 * PROGRAMA   : EJEMTABP 
000050 * OBJETIVO   : EJEMPLO DE TABLAS 
000060 * AUTOR      : JMP 
000070 * FECHA      : 23/08/94 
000080 * 
000090 ************************************************************************ 
000100 * 
000110 DEFINE DATA LOCAL 
000120 * 
000130    1 W-TABLA  (5)              /*   DEFINE UNA TABLA DE 5 OCURRENCIAS 
000140      2 W-ELEMENTOS (A10)       /*   DEFINICION DE CADA OCURRENCIA 
000150        INIT (1) <'A1A2A3A4A5'> /*   VALOR INICIAL DE LA OCURRENCIA 1 
000160             (2) <'B1B2B3B4B5'> /*   VALOR INICIAL DE LA OCURRENCIA 2 
000170             (3) <'C1C2C3C4C5'> /*   VALOR INICIAL DE LA OCURRENCIA 3 
000180             (4) <'D1D2D3D4D5'> /*   VALOR INICIAL DE LA OCURRENCIA 4 
000190             (5) <'E1E2E3E4E5'> /*   VALOR INICIAL DE LA OCURRENCIA 5 
000200    1 REDEFINE W-TABLA 
000210      2 W-TIPO-REGISTRO (1:5) 
000220        3 W-TIPO-1      (A2) 
000230        3 W-TIPO-2      (A2) 
000240        3 W-TIPO-3      (A2) 
000250        3 W-TIPO-4      (A2) 
000260        3 W-TIPO-5      (A2) 
000270 * 
000280    1 X     (N3) 
000290 * 
000300 END-DEFINE 
000310 ********************************************************************** 
000320 * 
000330     FOR X = 1 TO 5 
000340         DISPLAY  W-ELEMENTOS     (X) 
000350                  W-TIPO-REGISTRO (X) 
000360          '/T1'   W-TIPO-1  (X) 
000370          '/T2'   W-TIPO-2  (X) 
000380          '/T3'   W-TIPO-3  (X) 
000390          '/T4'   W-TIPO-4  (X) 
000400          '/T5'   W-TIPO-5  (X) 
000410     END-FOR 
000420     END 
****** **************************** bottom of data **************************** 
 
Enter-PF13--PF14--PF15--PF16--PF17--PF18--PF19--PF20--PF21--PF22--PF23--PF24--- 
      Help  Split End   nat M Rfind NAT L Up    Down  Swap  Left  Right Retri 
******************************************************************************* 
Page      1                                                  94-10-09  10:05:13 
 
W-ELEMENTOS               W-TIPO-REGISTRO 
                                                         T1 T2 T3 T4 T5 
            W-TIPO-1 W-TIPO-2 W-TIPO-3 W-TIPO-4 W-TIPO-5 
----------- -------- -------- -------- -------- -------- -- -- -- -- -- 
 
A1A2A3A4A5  A1       A2       A3       A4       A5       A1 A2 A3 A4 A5 
B1B2B3B4B5  B1       B2       B3       B4       B5       B1 B2 B3 B4 B5 
C1C2C3C4C5  C1       C2       C3       C4       C5       C1 C2 C3 C4 C5 
D1D2D3D4D5  D1       D2       D3       D4       D5       D1 D2 D3 D4 D5 
E1E2E3E4E5  E1       E2       E3       E4       E5       E1 E2 E3 E4 E5 
 
******************************************************************************* 
� 
 


